# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: Armin 'IsSuE' Praher [issue.at.gmx.dot.at]
# Contributor: [vEX] <niechift.dot.vex.at.gmail.dot.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>

pkgname=teeworlds
pkgver=0.6.2
pkgrel=1
pkgdesc="A multiplayer 2D shooter"
arch=('i686' 'x86_64')
url="https://www.teeworlds.com"
license=('custom')
depends=('alsa-lib' 'glu' 'sdl' 'freetype2')
makedepends=('python' 'bam' 'mesa')
replaces=('teewars')
source=(https://www.teeworlds.com/files/${pkgname}-${pkgver}-source.tar.gz
        teeworlds.desktop
		teeworlds.png)
md5sums=('649f6c05a7b117096b238c07062edc39'
         '3944aaf7a8c19a83510b2b39c4cddf5f'
         '44f39e398409232a0e0a431ed46d8cc8')

build() {
	# Build teeworlds
	cd ${srcdir}/${pkgname}-*-source

	bam server_release client_release
}

package() {
	cd ${srcdir}/${pkgname}-*-source

	# Install data files
	mkdir -p ${pkgdir}/usr/share/${pkgname}/data
	cp -r data/* ${pkgdir}/usr/share/${pkgname}/data

	install -Dm755 ${pkgname} ${pkgdir}/usr/bin/${pkgname}
	install -Dm755 ${pkgname}_srv ${pkgdir}/usr/bin/${pkgname}_srv

	install -Dm644 ${srcdir}/${pkgname}.desktop \
		${pkgdir}/usr/share/applications/${pkgname}.desktop
	install -Dm644 ${srcdir}/${pkgname}.png \
		${pkgdir}/usr/share/pixmaps/${pkgname}.png

	install -Dm644 license.txt \
		${pkgdir}/usr/share/licenses/${pkgname}/license.txt
}
